
HT48 & HT46 LCM Interface Design 

D/N：HA0013E 

Introduction 
The following note introduces a way of using 8 bit MCUs to control a DV16100NRB LCD 

display driver. The LCM is driven and controlled by a built-in Hitachi HD44780. The 

emphasis will lie on how to make the MCU produce the correct signals to meet the LCM 

timing requirements. For time order and instrction details, please refer to the LCM data 

sheets. The LCM can work with a 4 bit or 8 bit mode. In the 4 bit mode, it takes two 

transmission periods to send one byte or one instruction. In the 8 bit mode, it takes only 

one single transmission period to send one byte however it requires 4 more I/O pins. For 

the coding type, examples of using the ＃define, if, else and endif can be seen in the 

program. 

Circuit 
Set PB0~PB7 as I/O ports for data transmission, and PC0~PC2 as output ports for LCM 

control. Ports can be re-configured in order to meet the usersí need.  

Program Example 
;File name: lcm.asm 
;writer：Chris 
;explanation：main program of LCM control 
;============================================================== 
;choose LCM mode 
;four_bit＝0,LCM work in 8 bit mode,four_bit=1 LCM work in 4 bit mode 
;============================================================== 
;#define four_bit 0  ;LCM work in 8 mode 
#define four_bit 1  ;LCM work in 4 mode 
;============================================================== 
;the next part cannot be modified 
;============================================================== 
#include ht48r30a-1.inc 
;-------------------------------------------------------------- 
;LCM.INC for DV-16100NRB 
LCM_CLS  EQU 01H 
CURSOR_HOME EQU 02H 
CURSOR_SR  EQU 14H 
CURSOR_SL  EQU 10H 
INCDD_CG_SHF_C EQU 06H 
TURN_ON_DISP EQU 0FH 
LCD_ON_CSR_OFF EQU 0CH 
;-------------------------------------------------------------- 
lcm_data  equ pb ;define PORT B as data port 
lcm_data_ctrl equ pbc 
lcm_ctrl  equ pc ;define PORT C as LCM control port 
ctrl_ctrl  equ pcc 
 
extern busy_chk:near 
extern delay:near 
extern write_char:near  
extern snd_cmd:near 
 
;LCM Display Commands and control Signal name. 
E equ 0 
RW equ 1 
RS equ 2 
;-------------------------------------------------------------- 
data  .section ′data′ 
coun0  db ? 
coun1  db ? 
coun2  db ? 
msg  db ? 
tmp  db ? 
;-------------------------------------------------------------- 

code .section  at 0 ′code′ ;program 
 ORG 00h   ;set ISR address 
 jmp start 
start:     ;program start 
 clr lcm_data_ctrl ;setup LCM data port as outputs 
 clr ctrl_ctrl  ;setup LCM control port as outputs 
 clr lcm_data  ;clear LCM data port to 0 
 clr lcm_ctrl  ;clear LCM control port to 0 
DISPLAY_INIT: 
LCM_DELAY:    ;loop delay program for LCD control 
 mov a,0ffh 
 mov coun1,a 
 mov coun0,a 
lp0: 
 sdz coun1 
 jmp lp0 
 sdz coun0 
 jmp lp0 
 
 If four_bit 
  mov a,28h  ;4 bit mode 
 else 
  mov a,38h  ;8 bit mode 
 endif 
 call snd_cmd  ;execute write-in 
 MOV  A, 04H 
 CALL DELAY  
 
CMD_SEQ: 
 If four_bit 
  mov a,28h  ;4 bit mode 
 else 
  mov a,38h  ;8 bit mode 
 endif 
 call snd_cmd  ;execute write-in 
 MOV  A, 04H 
 CALL DELAY  
 
 if four_bit 
  mov a,80h 
 else 
  mov a, 38h 
 endif 
 call snd_cmd  ;execute write-in 
 MOV  A, 04H 
 CALL DELAY 
 
 call busy_chk ;CALL ″busy″ check program 
 mov  a,lcm_cls ;define ACC to LCM_CLS 

 call snd_cmd  ;CALL ″write-in″ program 
 MOV  A, 04H 
 CALL DELAY 
 
 call busy_chk 
 mov  a,TURN_ON_DISP ;define ACC to TURN_ON_DISP 
 call snd_cmd 
 MOV  A, 04H 
 CALL DELAY 
 
 call busy_chk 
 mov  a,INCDD_CG_SHF_C ;define ACC to INCDD_CG_SHF_C 
 call snd_cmd   
 MOV  A, 04H 
 CALL DELAY 

;-------------------------------------------------------------- 

 ;inc addr of DD ram & shift 
 ;the cursor to the right at 
 ;the time of write to DD/CG 
 ;RAM. 
;-------------------------------------------------------------- 
loop: 
 call busy_chk 
 mov  a,lcm_cls 
 call snd_cmd 
 MOV  A, 04H 
 CALL DELAY 
 
 call busy_chk 
 mov a,cursor_home ;define ACC to cursor_home 
 call snd_cmd 
 MOV  A, 04H 
 CALL DELAY 
 
 Call busy_chk  
 Clr  tblp  ;clear look up pointer to 0 
 
;initial display ″HOLTEK 8 bit uC″ 
agn: 
 tabrdl msg  ;define msg to data from the last page 
      
 mov  a,msg 
 mov  tmp,a 
 mov  a,24h 
 xorm a,msg 
 sz  msg  ;check if msg shows 0 
 jmp  agn1  ;if msg≠0, jump to agn1 
 jmp  secn_line ;if msg＝0, 
;jump to secn_line. 

agn1: 
 call busy_chk 
 mov  a,tmp 
 call write_char ;CALL ″write-in″ program 
 Inc  tblp 
 Jmp  agn 
secn_line: 
 inc  tblp  ;add 1 to look up pointer 
 call busy_chk 
 mov  a,0c0h  ;second row address 
 call snd_cmd 
 call busy_chk 
snd_line: 
 tabrdl msg 
 mov  a,msg 
 mov tmp,a 
 mov  a,24h 
 xorm a,msg 
 sz  msg 
 jmp  snd_lin1 
 mov  a,lcd_on_csr_off ;define ACC to lcd_on_csr_off 
 call snd_cmd 
 jmp  lp 
snd_lin1: 
 mov  a,tmp 
 call write_char 
 call busy_chk 
 inc  tblp  ;add 1 to look up pointer 
 jmp  snd_line 
lp: 
; jmp  start 
 jmp  lp 

;-------------------------------------------------------------- 

holtek_tbl .section at 700h ′code′ ;table at last page 
htk_tbl: 
; dc 4800h,4f00h,4c00h,5400h,4500h,4b00h,2000h,3800h,2400h 
; dc 2000h,6200h,6900h,7400h,2000h,7500h,4300h,2400h 
 
; dc 0048h,004fh,004ch,0054h,0045h,004bh,0020h,0038h,0024h 
; dc 0020h,0062h,0069h,0074h,0020h,0075h,0043h,0024h 
;note：the following parts is my correction 
 dc 0057h,0075h,0020h,004ah,0069h,006eh,0067h,0020h,0024h 
 dc 0020h,004ah,0069h,006eh,0067h,0020h,0020h,0020h,0024h 
 
 end 

;File name: lcmglob.inc 
;writer：Chris 
;explanation：define all constants 
lcm_data  equ pb 
lcm_data_ctrl equ pbc 
lcm_ctrl  equ pc 
ctrl_ctrl  equ pcc 
E   equ 0 
RW   equ 1 
RS   equ 2 
 
 
;File name: command.asm 
;writer：Chris 
;explanation: use LCM control instruction document as an interface  
;of the external main program LCM.ASM. Add it with the main program 
;to the project 
;============================================================== 
;choose LCM mode 
;four_bit＝0,LCM work in 8 bit mode,four_bit=1 LCM work in 4 bit mode 
;============================================================== 
;#define four_bit 0  ;LCM work in 8 bit mode 
#define four_bit 1  ;LCM work in 4 bit mode 
;============================================================== 
;the following part cannot be modified 
;============================================================== 
;LCM command module begins here 
include ht48r30a-1.inc 
include lcmglob.inc 
public busy_chk 
public delay 
public write_char 
public snd_cmd 
;-------------------------------------------------------------- 
scode .section ′code′  ;write LCM control instruction 
     ;program 
snd_cmd: 
 if four_bit  ;if in 4 bit mode 
 mov  dtmp,a 
 and  a,0f0h  ;input high nibble 4 bit to ACC 
 endif 
 
  mov  lcm_data,a ;write LCM control instruction into 
      ;data port 
 clr  lcm_ctrl.rw ;set write-in mode 
 Clr  lcm_ctrl.rs 
 set  lcm_ctrl.e 
 clr  lcm_ctrl.e ;write instruction to LCM 

 if four_bit 
 swapa dtmp 
 and  a,0f0h  ;input low nibble 4 bit to ACC. 
 mov  lcm_data,a 
 set  lcm_ctrl.e 
 clr  lcm_ctrl.e 
 endif 
 ret 
acode .section ′code′ 
busy_chk: 
 set  lcm_data_ctrl ;set LCM data port as input status 
 clr  lcm_ctrl.rs ;set LCM as test; in BUSY mode 
 set  lcm_ctrl.rw 
 set  lcm_ctrl.e 
 mov  a,lcm_data ;define ACC to LCM data port 
 if four_bit 
  and a,0f0h ;read high nibble to ACC in 4 bit mode 
  mov dtmp,a  
  set lcm_ctrl.e  
  swapa lcm_data  
  clr lcm_ctrl.e  
  and a,0fh  ;read low nibble into ACC 
  Or a,dtmp ;make 8 bit 
 endif 
 clr lcm_ctrl.e 
 sz acc.7  ;check if LCM is in busy status 
 jmp busy_chk  ;if yes, return to busy signal 
 clr lcm_ctrl.rw 
 clr lcm_data_ctrl ;if not, reset LCD data port as 
     ;output status 
 ret 
;-------------------------------------------------------------- 
wcode .section ′code′ 
write_char: 
 if four_bit 
  mov dtmp,a 
  and a,0f0h 
 endif 
  mov lcm_data,a 
  clr lcm_ctrl.rw 
  set lcm_ctrl.rs 
  set lcm_ctrl.e 
  clr lcm_ctrl.e 
 
 if four_bit 
  swapa dtmp 
  and a,0f0h 
  mov lcm_data,a 
  set lcm_ctrl.e 

  clr lcm_ctrl.e 
 endif 
 ret 

;-------------------------------------------------------------- 

dsec .section ′data′ 
dtmp db ? 
dtmp2 db ? 

;-------------------------------------------------------------- 

code .section ′code′ 
delay: 
 mov  dtmp,a 
 set  dtmp2 
drep: 
 sdz  dtmp2 
 jmp  drep 
 sdz  dtmp 
 jmp  drep 
 
 ret 

Program Instruction 
The program firstly executes the include section. Define PORT B as LCM data port and 

PORT C as LCM control ports. The LCD program can be compiled in 4 bit or 8 bit mode 

through define instructions. The LCM will reset automatically after power on yet most 

LCM programs can be initialized using software. In this application the initial program will 

go to ″START″. In accordance with the HD44780 specifications, each instruction takes at 

least 4.5ms to execute. This is the function of the LCM_DELAY program. Before the LCM 

has fully initialized, there is no way to check if LCM is in the BUSY status. Instructions to 

the LCM must depend upon the code definition of the HD44780. LCM.INC has defined 

some common instructions. Before writing instructions to the LCM, check if the LCM is in 

BUSY status. This is the function of the BUSY_CHK program. Confirm that LCM is not in 

the BUSY status before transmitting data to the LCM. Place the ASCII code at the last 

page, save the data to the ACC by a look up process and show the code through the 

WRITE_CHAR program of the LCM. 



                      HT48 & HT46 LCM Interface Design 

 

 9

RS, R/W and E signal Truth Table： 

